
 Instruction List Scripts

OpIntfc
(* 
  Description: Example 13.4 - Operator interface For tank control 

  Copyright (c) 2023 Dogwood Valley Press, LLC 
*) 
(* Check for T428_Min less than min or greater than max *) 

LT_REAL( T428_Min, 2.1) 
        JMPCN TUpr 

ADD_REAL(0.0, 2.1) 
        ST_REAL T428_Min 
TUpr:    GT_REAL( T428_Min, 13.4) 
        JMPCN DoAdd 

ADD_REAL(0.0, 13.4) 
        ST_REAL T428_Min 
(* Max level is min level +1.5 *) 
DoAdd:    ADD_REAL( T428_Min, 1.5) 
        ST_REAL    T428_Max 
(* Low level lamp when level < 4 feet *) 

LT_REAL( LT428_Val, 4.0) 
        ST_BOOL    T428_LOLA 
(* Level below 2.0 activates horn *) 

LT_REAL( LT428_Val, 2.0) 
        ST_BOOL    T428_Hrn_Act 

PT        Hrn_Ons 
(* Detect Ack transition *) 

LD_BOOL Alm_Ack 
PT      Ack_Ons 
ST_BOOL AckTrig 

(* Horn activated when level drops below 2.0 or has 
   remained below 2.0 for 5 minutes. Acknowledge button 
   silences it. *) 

LD_BOOL    Hrn_Ons 
OR        Ack_Tmr_Dn 
OR( 

        LD_BOOL    T428_HORN 
ANDN    AckTrig 
) 

        ST_BOOL    T428_HORN 
(* Time 5 minutes level stays below 2.0 after horn ackd *) 

LD_BOOL    AckTrig 
OR        Ack_Tmr_En 
AND        T428_Hrn_Act 
ANDN    Ack_Tmr_Dn 
ST_BOOL    Ack_Tmr_En 
TMR_TENTHS(Ack_Tmr, 3000) 

        ST_BOOL    Ack_Tmr_Dn 

TankCtl
(* 
   Description: Example 13.3 - Tank Control 

   Copyright (c) 2015 Dogwood Valley Press, LLC 
*) 
(* Convert analog input to level    *) 

INT_TO_REAL(LT428_MEAS) 
        ST_REAL TmpR 

DIV_REAL( TmpR, 32000.0) 
        ST_REAL TmpR 

SUB_REAL( 15.0, 1.0) 
        ST_REAL TmpR2 

MUL_REAL(TmpR, TmpR2) 
        ST_REAL TmpR 

ADD_REAL( TmpR, 1.0) 
        ST_REAL LT428_Val 
(* Comparisons for low and high levels *) 

LT_REAL(LT428_Val,T428_Min) 
        ST_BOOL T428_Low 

GT_REAL(LT428_Val, T428_Max) 
        ST_BOOL T428_High 
(* Control of tank level *) 

LD_BOOL T428_Low 
OR         XV427_OPEN 
ANDN    T428_High 
AND        T428_Cntrl 
ST_BOOL    XV427_OPEN                

